
  Usage


Cloak.Ecto helps you create Ecto.Type modules which automatically encrypt
and decrypt your data. You simply define a type and set the type of your fields, and
Cloak.Ecto handles the rest.
defmodule MyApp.Encrypted.Binary do
  use Cloak.Ecto.Binary, vault: MyApp.Vault
end
defmodule MyApp.EctoSchema do
  use Ecto.Schema

  schema "table_name" do
    field :encrypted_field, MyApp.Encrypted.Binary

    # ...
  end
end
When Ecto writes the fields to the database, Cloak encrypts the values into a
binary blob, using a configured encryption algorithm chosen by you.
iex> Repo.insert!(%MyApp.EctoSchema{encrypted_field: "plaintext"})
08:46:08.862 [debug] QUERY OK db=3.4ms
INSERT INTO "table_name" ("encrypted_field") 
VALUES ($1) RETURNING "id", "encrypted_field" [
  <<1,10, 65, 69, 83, 46, 67, 84, 82, 46, 86, 49, 
    69, 92, 173, 219, 203, 238, 26, 58, 236, 5, 
    104, 23, 12, 10, 182, 31, 221, 89, 22, 58, 
    34, 79, 109, 30, 70, 254, 56, 93, 102, 84>>
]
Likewise, when Ecto reads the encrypted blob out of the database, Cloak will
automatically decrypt the value into the intended data type at runtime.
iex> Repo.get(MyApp.EctoSchema, 1)
%MyApp.EctoSchema{encrypted_field: "plaintext"}
For complete usage instructions, see the Hex documentation.

  
    
  
  Troubleshooting


See our troubleshooting guide for solutions to common issues.

  
    
  
  Notable Features


	Transparent, easy to use encryption for database fields
	Fully compatible with umbrella projects
	Bring your own encryption algorithm, if you want
	Mix task for key rotation: mix cloak.migrate


  
    
  
  Security Notes


	 Supported Algorithms: Cloak's built-in encryption modules rely on Erlang's 
:crypto module. Cloak supports the following algorithms out of the box:
	AES.GCM
	AES.CTR


	Encrypted Data Not Searchable: Cloak uses random IVs for each ciphertext. This 
means that the same value will not encrypt to the same value twice. As a result,
encrypted columns are not queryable. However, Cloak does provide easy ways to
create hashed, searchable columns.

	Runtime Data is not Encrypted: Cloak encrypts data at rest in the database. 
The data in your Ecto structs at runtime is not encrypted.

	No Support for User-specific Encryption Keys: Cloak's Ecto.Type modules do not
support user-specific encryption keys, due to limitations on the Ecto.Type 
behaviour. However, you can still use Cloak's ciphers to implement these in your 
application logic.



  
    
  
  Use Without Ecto


If you want to use Cloak without Ecto, see
cloak instead.

  
    
  
  Local Development


To develop this library locally, you will need to install the correct version
of Elixir and Postgres. The easiest way to set everything up is with Docker
and docker-compose:
$ cd cloak_ecto
# Runs the bin/test script in the context of Docker
$ docker-compose run code bin/test
# To access a terminal with mix, use this command:
$ docker-compose run code bash
# Run any command of your choosing:
root@234098234oij:/app# mix docs




  

    
Installing Cloak.Ecto
    

This guide will walk you through installing Cloak.Ecto in your project.

  
    
  
  Add the Dependency


First, add :cloak_ecto to your dependencies in mix.exs:
{:cloak_ecto, "~> 1.2.0"}
Run mix deps.get to fetch the dependency. Since :cloak_ecto relies
on :cloak, the :cloak package will also be installed.

  
    
  
  Generate a Key


You'll need a secret key for encryption. This is easy to generate in the
IEx console.
$ iex
iex> 32 |> :crypto.strong_rand_bytes() |> Base.encode64()
"aJ7HcM24BcyiwsAvRsa3EG3jcvaFWooyQJ+91OO7bRU="

This will generate a relatively strong encryption 256-bit encryption
key encoded with Base64.

  
    
  
  Create a Vault


Next, create a Cloak.Vault for your project.
defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app
end
Configure it as shown in the Cloak.Vault documentation, with at least one
active cipher. Note that the :key needs to be decoded from Base64 encoding into
its raw binary form.
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: Base.decode64!("your-key-here")}
  ]
If you want to fetch keys from system vars, you should use the init/1 callback
to configure the vault instead:
# Assumes that you have a CLOAK_KEY environment variable containing a key in
# Base64 encoding.
#
# export CLOAK_KEY="A7x+qcFD9yeRfl3GohiOFZM5bNCdHNu27B0Ozv8X4dE="

defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl GenServer
  def init(config) do
    config =
      Keyword.put(config, :ciphers, [
        default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: decode_env!("CLOAK_KEY")}
      ])

    {:ok, config}
  end

  defp decode_env!(var) do
    var
    |> System.get_env()
    |> Base.decode64!()
  end
end
Finally, add your vault to your supervision tree.
children = [
  MyApp.Vault
]

  
    
  
  Create Local Ecto Types


For each type of data you want to encrypt, define a local Ecto type:
defmodule MyApp.Encrypted.Binary do
  use Cloak.Ecto.Binary, vault: MyApp.Vault
end
You can find a complete list of available types in the "MODULES" documentation.

  
    
  
  Create Your Schema


If you want to encrypt an existing schema, see the guide on Encrypting
Existing Data.
If you're starting from scratch with a new Ecto.Schema, generate your
fields with the type :binary:
create table(:users) do
  add :email, :binary
  add :email_hash, :binary # will be used for searching
  # ...

  timestamps()
end
Your schema module should look like this:
defmodule MyApp.Accounts.User do
  use Ecto.Schema

  import Ecto.Changeset

  schema "users" do
    field :email, MyApp.Encrypted.Binary
    field :email_hash, MyApp.Hashed.HMAC
    # ... other fields

    timestamps()
  end

  @doc false
  def changeset(struct, attrs \\ %{}) do
    struct
    |> cast(attrs, [:email])
    |> put_hashed_fields()
  end

  defp put_hashed_fields(changeset) do
    changeset
    |> put_change(:email_hash, get_field(changeset, :email))
  end
end
This example also shows how you would make a given field queryable by
creating a mirrored _hash field. See Cloak.Ecto.HMAC for more details.

  
    
  
  Usage


Your encrypted fields will be transparently encrypted and decrypted as
data are loaded from the database.
Repo.get(Accounts.User, 1)
# => %Accounts.User{email: "test@example.com", email_hash: <<115, 6, 45, 135, 41, ...>>}
You can query by the mirrored _hash fields:
Repo.get_by(Accounts.User, email_hash: "test@example.com")
# => %Accounts.User{email: "test@example.com", ...}
And you're done! Cloak.Ecto is successfully installed.



  

    
How to Generate Encryption Keys
    

The easiest way to generate a key is via IEx.
$ iex
iex> 32 |> :crypto.strong_rand_bytes() |> Base.encode64()
"HXCdm5z61eNgUpnXObJRv94k3JnKSrnfwppyb60nz6w="

This will generate a strong 256-bit key encoded with Base64.



  

    
How to Encrypt Existing Data
    

When you adopt Cloak in an existing project, you often need to encrypt fields
that were not previously encrypted. This guide will help you through the
process, step by step.

  
    
  
  Create a Vault


Make sure you've set up (and configured!) a vault as described in the
Cloak.Vault documentation and the installation guide.

  
    
  
  Create Local Ecto Types


Create local Ecto types for each kind of field you want to encrypt, as
described in the Cloak.Vault documentation and the installation
guide.

  
    
  
  Add encrypted_* Fields


You'll need to create a duplicate version of each field you want to encrypt,
perhaps with the encrypted_ prefix.
alter table(:users) do
  add :encrypted_name, :binary
  add :encrypted_metadata, :binary
end
Then, add the new fields to your schema. Make sure to keep them up to
date with changes using a temporary put_encrypted_fields function in
your changeset.
defmodule MyApp.Accounts.User do
  use Ecto.Schema

  import Ecto.Changeset

  schema "users" do
    field :name, :string
    field :encrypted_name, MyApp.Encrypted.Binary
    field :metadata, :map
    field :encrypted_metadata, MyApp.Encrypted.Map
  end

  @doc false
  def changeset(struct, attrs \\ %{}) do
    struct
    |> cast(attrs, [:name, :metadata])
    |> put_encrypted_fields()
  end

  # Temporary function during the migration process, to ensure
  # that all changes are copied over to the new fields
  defp put_encrypted_fields(changeset) do
    changeset
    |> put_change(:encrypted_name, get_field(changeset, :name))
    |> put_change(:encrypted_metadata, get_field(changeset, :metadata))
  end
end

  
    
  
  Copy Data to New Fields


In a custom mix task or the IEx console, run the following:
MyApp.Accounts.User
|> MyApp.Repo.all()
|> Enum.map(fn user ->
  user
  |> Ecto.Changeset.change(%{
       encrypted_name: user.name,
       encrypted_metadata: user.metadata
     })
  |> MyApp.Repo.update!()
end)

  
    
  
  Remove Original Fields


Once you're confident that all the data has migrated successfully to the new
fields, you can write a migration to remove the old, unencrypted fields.
alter table(:users) do
  remove :name
  remove :metadata
end

rename table(:users), :encrypted_name, to: :name
rename table(:users), :encrypted_metadata, to: :metadata
You can also remove the temporary put_encrypted_fields/1 function, leaving
your schema like this:
defmodule MyApp.Accounts.User do
  use Ecto.Schema

  import Ecto.Changeset

  schema "users" do
    field :name, MyApp.Encrypted.Binary
    field :metadata, MyApp.Encrypted.Map
  end

  @doc false
  def changeset(struct, attrs \\ %{}) do
    struct
    |> cast(attrs, [:name, :metadata])
  end
end



  

    
How to Rotate Keys
    

Many organizations rotate their keys on a regular schedule for security
reasons. Cloak can do this with zero downtime.

  
    
  
  Add Your New Key


You will need to add your new key to your vault configuration, as the
first key in the :ciphers list. Demote and re-label existing keys.
For example, if your config currently looks like this:
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: <<...>>},
  ]
Then you should change the :default label to the new key, and demote the
existing key to the :retired label.
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V2", key: <<...>>},
    retired: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V1", key: <<...>>}
  ]
Labels don't matter, order does
The :ciphers configuration is order-dependent. The first key in the list
will be used as the default, regardless of its label. This guide uses the 
:default and :retired labels for clarity, but the order is what matters.


  
    
  
  Migrate Data To New Key


For each schema that uses your vault, run mix cloak.migrate:
mix cloak.migrate.ecto -r MyApp.Repo -s MyApp.Schema
Alternatively, you can migrate multiple schemas at once by configuring
the following values in config/config.exs:
config :my_app,
  cloak_repo: MyApp.Repo,
  cloak_schemas: [...]
With that in place, you can simply run:
mix cloak.migrate.ecto

  
    
  
  Remove Retired Key


Once you are confident that all data have been converted over to use
the new key, you may remove the :retired key from your configuration.
config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "AES.GCM.V2", key: <<...>>},
  ]
How It Works
Every ciphertext generated by Cloak is tagged with the :tag of the
configured cipher that produced it. 
As a result, Cloak will be able to find and use the correct cipher to decrypt
any old ciphertext as long as you leave its cipher configured.
When you add a new default key, Cloak will immediately begin encrypting
new data with the new key. Every time existing data are fetched from the
database, they will be decrypted with the old key, and if changed, will
be encrypted with the new key.
Even if you don't run mix cloak.migrate.ecto, data will gradually be
converted over to the new key as they are used. The migration task simply
speeds up the process.



  

    
Troubleshooting
    


  
    
  
  Error when validating default values


If you have a field in your schema with a :default value, and are running Ecto
3.6 or later, you may encounter an error like this:
** (ArgumentError) errors were found at the given arguments:

  * 1st argument: the table identifier does not refer to an existing ETS table

    (stdlib 4.0) :ets.lookup(MyApp.Vault.Config, :config)
    (cloak 1.1.2) lib/cloak/vault.ex:285: Cloak.Vault.read_config/1
    lib/cloak/vault.ex:224: MyApp.Vault.encrypt/1
    lib/cloak_ecto/type.ex:37: Cloak.Ecto.Encrypted.Binary.dump/1
    (ecto 3.8.4) lib/ecto/schema.ex:2198: Ecto.Schema.validate_default!/3
    (ecto 3.8.4) lib/ecto/schema.ex:1919: Ecto.Schema.__field__/4
    lib/my_app/post.ex:9: (module)
This error happens because cloak relies on an ETS table which is only available
at runtime, and Ecto 3.6+ tries to validate default values at compile time.
To fix it, add the :skip_default_validation option:
field :name, MyApp.Encrypted.Binary, default: "foo", skip_default_validation: true



  

    
Closure wrapping
    

Use the closure: true option in local Ecto type definitions to wrap decrypted
values in a zero-arity closure. This prevents the plaintext from accidentally
leaking in some contexts, such as in stacktraces, Inspect documents and JSON
serializations.
defmodule MyApp.Encrypted.WrappedBinary do
  use Cloak.Ecto.Binary, vault: MyApp.Vault, closure: true
end
In those places where the plaintext value is needed, unwrap the value by
invoking the closure as an anonymous function. Consider using
Plug.Crypto.prune_args_from_stacktrace/1
in functions that unwrap the plaintext, to prevent leakage in exceptions that
may occur further down the call stack:
def basic_auth(req, client) do
  # Unwrap string value from password closure
  HTTPClient.basic_auth(req, client.username, client.password.())
rescue
  e ->
    reraise e, Plug.Crypto.prune_args_from_stacktrace(System.stacktrace())
end



  

    
Troubleshooting
    


  
    
  
  Error when validating default values


If you have a field in your schema with a :default value, and are running Ecto
3.6 or later, you may encounter an error like this:
** (ArgumentError) errors were found at the given arguments:

  * 1st argument: the table identifier does not refer to an existing ETS table

    (stdlib 4.0) :ets.lookup(MyApp.Vault.Config, :config)
    (cloak 1.1.2) lib/cloak/vault.ex:285: Cloak.Vault.read_config/1
    lib/cloak/vault.ex:224: MyApp.Vault.encrypt/1
    lib/cloak_ecto/type.ex:37: Cloak.Ecto.Encrypted.Binary.dump/1
    (ecto 3.8.4) lib/ecto/schema.ex:2198: Ecto.Schema.validate_default!/3
    (ecto 3.8.4) lib/ecto/schema.ex:1919: Ecto.Schema.__field__/4
    lib/my_app/post.ex:9: (module)
This error happens because cloak relies on an ETS table which is only available
at runtime, and Ecto 3.6+ tries to validate default values at compile time.
To fix it, add the :skip_default_validation option:
field :name, MyApp.Encrypted.Binary, default: "foo", skip_default_validation: true



  

    
How To Upgrade From 0.9.x to 1.0.x
    

Cloak 1.0.0 separated the Ecto components into a separate package:
cloak_ecto. This required some breaking name changes.
	Cloak's default JSON library is now Jason, not Poison
	All Ecto-related modules now have the prefix Cloak.Ecto.
	Cloak's Ecto types have a new naming scheme: Cloak.Fields.Binary became
Cloak.Ecto.Binary.
	mix cloak.migrate has been renamed to mix cloak.migrate.ecto.


  
    
  
  Install cloak_ecto 1.0.0


Replace your cloak dependency with cloak_ecto.
{:cloak_ecto, "1.0.0"}

  
    
  
  Configure JSON Library


If you have not specified the :json_library setting in your Vault, and want
to continue using Poison, you will need to add that setting:
config :my_app, MyApp.Vault,
  json_library: Poison,
  # ...
Otherwise, you will need to add Jason to your dependencies:
{:jason, "~> 1.0"}

  
    
  
  Update Cloak Field References


Update all your Cloak field references from the old naming scheme to the new.
For example:
use Cloak.Fields.Binary, vault: MyApp.Vault
Should now become:
use Cloak.Ecto.Binary, vault: MyApp.Vault

  
    
  
  Update Migration References


If you use Cloak.Migrator directly, rename your references to
Cloak.Ecto.Migrator instead. If you use mix cloak.migrate, rename any
references to mix cloak.migrate.ecto instead..



  

    
Cloak.Ecto.CustomCursor behaviour
    



      
Defines a list of fields to use as a cursor when migrating a schema's table
to a new encryption key using mix cloak.migrate.ecto.

  
    
  
  When to Use This


This behaviour is useful in rare situations where your primary key cannot
be used as a cursor to page through the table. You need this if primary key
cannot meet the "Column Requirements" below.
For example, you don't need to use this module if your primary key is:
	An integer
	A PostgreSQL UUID
	A MongoDB ObjectID
	Your primary key already meets the "Column Requirements"


  
    
  
  Column Requirements


Each column you specify for the cursor must meet the following
requirements:
	Each column must contain sortable values
	Each column must be comparable using the > operator
	The combined value of the columns must be unique in the table


  
    
  
  Example


defmodule MyApp.MySchema do
  use Ecto.Schema

  @behaviour Cloak.Ecto.CustomCursor

  schema "table" do
    # ...
  end

  @impl Cloak.CustomCursor
  def __cloak_cursor_fields__ do
    [:sequential_id, :inserted_at]
  end
end

      


      
        Summary


  
    Callbacks
  


    
      
        __cloak_cursor_fields__()

      


        Returns a list of cursor fields.



    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    __cloak_cursor_fields__()


      
       
       View Source
     


  


  

      

          @callback __cloak_cursor_fields__() :: [atom()]


      


Returns a list of cursor fields.

  
    
  
  Example


@impl Cloak.Ecto.CustomCursor
def __cloak_cursor_fields__ do
  [:sequential_id, :inserted_at]
end

  


        

      



  

    
Cloak.Ecto.Binary 
    



      
An Ecto.Type to encrypt a binary field.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Binary module in your project:
defmodule MyApp.Encrypted.Binary do
  use Cloak.Ecto.Binary, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Binary
end

      





  

    
Cloak.Ecto.Date 
    



      
An Ecto.Type to encrypt Date fields.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Date module in your project:
defmodule MyApp.Encrypted.Date do
  use Cloak.Ecto.Date, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Date
end

      





  

    
Cloak.Ecto.DateTime 
    



      
An Ecto.Type to encrypt DateTime fields.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.DateTime module in your project:
defmodule MyApp.Encrypted.DateTime do
  use Cloak.Ecto.DateTime, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.DateTime
end

      





  

    
Cloak.Ecto.Decimal 
    



      
An Ecto.Type to encrypt a decimal field, relying on the Decimal
library, which is a dependency of Ecto.
Decimals are more precise than floats.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Decimal module in your project:
defmodule MyApp.Encrypted.Decimal do
  use Cloak.Ecto.Decimal, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Decimal
end

      





  

    
Cloak.Ecto.Float 
    



      
An Ecto.Type to encrypt a float field. Consider using Cloak.Ecto.Decimal
instead if precision is important.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Float module in your project:
defmodule MyApp.Encrypted.Float do
  use Cloak.Ecto.Float, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Float
end

      





  

    
Cloak.Ecto.HMAC behaviour
    



      
A custom Ecto.Type for hashing fields using :crypto.hmac/3.

  
    
  
  Why


If you store a hash of a field's value, you can then query on it as a proxy
for an encrypted field. This works because HMAC is deterministic and
always results in the same value, while secure encryption does not. Be
warned, however, that hashing will expose which fields have the same value,
because they will contain the same hash.

  
    
  
  Security


HMAC is more secure than Cloak.Ecto.SHA256, because it uses a
secret to obfuscate the hash. This makes it harder to guess the value of
the field.

  
    
  
  Configuration


Create an HMAC field in your project:
defmodule MyApp.Hashed.HMAC do
  use Cloak.Ecto.HMAC, otp_app: :my_app
end
Then, configure it with a :secret and :algorithm, either using
mix configuration:
config :my_app, MyApp.Hashed.HMAC,
  algorithm: :sha512,
  secret: "secret"
Or using the init/1 callback to fetch configuration at runtime:
defmodule MyApp.Hashed.HMAC do
  use Cloak.Ecto.HMAC, otp_app: :my_app

  @impl Cloak.Ecto.HMAC
  def init(config) do
    config = Keyword.merge(config, [
      algorithm: :sha512,
      secret: System.get_env("HMAC_SECRET")
    ])

    {:ok, config}
  end
end

  
    
  
  Usage


Create the hash field with the type :binary. Add it to your schema
definition like this:
schema "table" do
  field :field_name, MyApp.Encrypted.Binary
  field :field_name_hash, MyApp.Hashed.HMAC
end
Ensure that the hash is updated whenever the target field changes with the
put_change/3 function:
def changeset(struct, attrs \\ %{}) do
  struct
  |> cast(attrs, [:field_name, :field_name_hash])
  |> put_hashed_fields()
end

defp put_hashed_fields(changeset) do
  changeset
  |> put_change(:field_name_hash, get_field(changeset, :field_name))
end
Query the Repo using the :field_name_hash in any place you would typically
query by :field_name.
user = Repo.get_by(User, email_hash: "user@email.com")

      


      
        Summary


  
    Types
  


    
      
        algorithms()

      


        HMAC algorithms supported by Cloak.Field.HMAC



    





  
    Callbacks
  


    
      
        init(config)

      


        Configures the HMAC field using runtime information.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    algorithms()


      
       
       View Source
     


  


  

      

          @type algorithms() :: :md5 | :ripemd160 | :sha | :sha224 | :sha256 | :sha384 | :sha512


      


HMAC algorithms supported by Cloak.Field.HMAC

  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    init(config)


      
       
       View Source
     


  


  

      

          @callback init(config :: Keyword.t()) :: {:ok, Keyword.t()} | {:error, any()}


      


Configures the HMAC field using runtime information.

  
    
  
  Example


@impl Cloak.Fields.HMAC
def init(config) do
  config = Keyword.merge(config, [
    algorithm: :sha512,
    secret: System.get_env("HMAC_SECRET")
  ])

  {:ok, config}
end

  


        

      



  

    
Cloak.Ecto.Integer 
    



      
An Ecto.Type to encrypt integer fields.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Integer module in your project:
defmodule MyApp.Encrypted.Integer do
  use Cloak.Ecto.Integer, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Integer
end

      





  

    
Cloak.Ecto.IntegerList 
    



      
An Ecto.Type to encrypt a list of integers.

  
    
  
  Configuration


You can customize the json library used for for converting the lists.
config :my_app, MyApp.Vault,
  json_library: Jason

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.IntegerList module in your project:
defmodule MyApp.Encrypted.IntegerList do
  use Cloak.Ecto.IntegerList, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.IntegerList
end

      





  

    
Cloak.Ecto.Map 
    



      
An Ecto.Type to encrypt maps.

  
    
  
  Configuration


You can customize the json library used for for converting the maps to
and from JSON during encryption and decryption.
config :my_app, MyApp.Vault,
  json_library: Jason

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Map module in your project:
defmodule MyApp.Encrypted.Map do
  use Cloak.Ecto.Map, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Map
end
On encryption, the map will first be converted to JSON using the configured
:json_library, and then encrypted. On decryption, the :json_library
will be used to convert it back to a map.
This means that on decryption, atom keys will become string keys.
# ON WRITE
%{hello: "world"}

# ON READ
# Keys converted to strings
%{"hello" => "world"}

      





  

    
Cloak.Ecto.NaiveDateTime 
    



      
An Ecto.Type to encrypt NaiveDateTime fields.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.NaiveDateTime module in your project:
defmodule MyApp.Encrypted.NaiveDateTime do
  use Cloak.Ecto.NaiveDateTime, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.NaiveDateTime
end

      





  

    
Cloak.Ecto.PBKDF2 behaviour
    



      
A custom Ecto.Type for deriving a key for fields using
PBKDF2.
PBKDF2 is more secure than Cloak.Ecto.HMAC and
Cloak.Fields.SHA256 because it uses key
stretching to increase the
amount of time to compute hashes. This slows down brute-force attacks.

  
    
  
  Why


If you store a hash of a field's value, you can then query on it as a
proxy for an encrypted field. This works because PBKDF2 is deterministic
and always results in the same value, while secure encryption does not.
Be warned, however, that hashing will expose which fields have the same
value, because they will contain the same hash.

  
    
  
  Dependency


To use this field type, you must install the :pbkdf2 library in your
mix.exs file.
{:pbkdf2, "~> 2.0"}
If you are using Erlang >= 24, you will need to use a forked version,
because pbkdf2 version 2.0.0 uses :crypto.hmac functions that were
removed in Erlang 24.
{:pbkdf2, "~> 2.0", github: "miniclip/erlang-pbkdf2"}

  
    
  
  Configuration


Create a PBKDF2 field in your project:
defmodule MyApp.Hashed.PBKDF2 do
  use Cloak.Ecto.PBKDF2, otp_app: :my_app
end
Then, configure it with a :secret, an :algorithm, the maximum :size
of the stored key (in bytes), and a number of :iterations, either using
mix configuration:
config :my_app, MyApp.Hashed.PBKDF2,
  algorithm: :sha256,
  iterations: 600_000,
  secret: "secret",
  size: 64
Or using the init/1 callback to fetch configuration at runtime:
defmodule MyApp.Hashed.PBKDF2 do
  use Cloak.Ecto.PBKDF2, otp_app: :my_app

  @impl Cloak.Ecto.PBKDF2
  def init(config) do
    config = Keyword.merge(config, [
      algorithm: :sha256,
      iterations: 600_000,
      secret: System.get_env("PBKDF2_SECRET")
    ])

    {:ok, config}
  end
end

  
    
  
  Usage


Create the hash field with the type :binary. Add it to your schema
definition like this:
schema "table" do
  field :field_name, MyApp.Encrypted.Binary
  field :field_name_hash, MyApp.Hashed.PBKDF2
end
Ensure that the hash is updated whenever the target field changes with the
put_change/3 function:
def changeset(struct, attrs \\ %{}) do
  struct
  |> cast(attrs, [:field_name, :field_name_hash])
  |> put_hashed_fields()
end

defp put_hashed_fields(changeset) do
  changeset
  |> put_change(:field_name_hash, get_field(changeset, :field_name))
end
Query the Repo using the :field_name_hash in any place you would typically
query by :field_name.
user = Repo.get_by(User, email_hash: "user@email.com")

      


      
        Summary

          @type algorithms() ::
  :md4 | :md5 | :ripemd160 | :sha | :sha224 | :sha256 | :sha384 | :sha512


      


Digest algorithms supported by Cloak.Field.PBKDF2

  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    init(config)


      
       
       View Source
     


  


  

      

          @callback init(config :: Keyword.t()) :: {:ok, Keyword.t()} | {:error, any()}


      


Configures the PBKDF2 field using runtime information.

  
    
  
  Example


@impl Cloak.Ecto.PBKDF2
def init(config) do
  config = Keyword.merge(config, [
    algorithm: :sha256,
    secret: System.get_env("PBKDF2_SECRET")
  ])

  {:ok, config}
end

  


        

      



  

    
Cloak.Ecto.SHA256 
    



      
An Ecto.Type which hashes the field value using the SHA256 algorithm.

  
    
  
  Why


If you store a hash of a field's value, you can then query on it as a proxy
for the encrypted field. This works because SHA256 is deterministic and
always results in the same value, while secure encryption does not. Be
warned, however, that hashing will expose which fields have the same value,
because they will contain the same hash.

  
    
  
  Security


For a more secure hashing method, see one of the following alternatives:
	Cloak.Ecto.HMAC
	Cloak.Ecto.PBKDF2


  
    
  
  Usage


Create the hash field with the type :binary. Add it to your schema
definition like this:
schema "table" do
  field :field_name, MyApp.Encrypted.Binary
  field :field_name_hash, Cloak.Ecto.SHA256
end
Ensure that the hash is updated whenever the target field changes with the
put_change/3 function:
def changeset(struct, attrs \\ %{}) do
  struct
  |> cast(attrs, [:field_name, :field_name_hash])
  |> put_change(:field_name_hash, get_field(changeset, :field_name))
end
Query the Repo using the :field_name_hash in any place you would typically
query by :field_name.
user = Repo.get_by(User, email_hash: "user@email.com")

      





  

    
Cloak.Ecto.StringList 
    



      
An Ecto.Type to encrypt a list of strings.

  
    
  
  Configuration


You can customize the json library used for for converting lists.
config :my_app, MyApp.Vault,
  json_library: Jason

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.StringList module in your project:
defmodule MyApp.Encrypted.StringList do
  use Cloak.Ecto.StringList, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.StringList
end
On encryption, the list will first be converted to JSON using the configured
:json_library, and then encrypted. On decryption, the :json_library will
be used to convert it back to a list of strings.

      





  

    
Cloak.Ecto.Time 
    



      
An Ecto.Type to encrypt Time fields.

  
    
  
  Migration


The database field must be of type :binary.
add :encrypted_field, :binary

  
    
  
  Usage


Define an Encrypted.Time module in your project:
defmodule MyApp.Encrypted.Time do
  use Cloak.Ecto.Time, vault: MyApp.Vault
end
Then, define the type of your desired fields:
schema "table_name" do
  field :encrypted_field, MyApp.Encrypted.Time
end

      





  

    
mix cloak.migrate.ecto 
    



      
Migrates a schema table to a new encryption cipher.

  
    
  
  Rationale


Cloak vaults will automatically decrypt fields which were encrypted
by a retired key, and reencrypt them with the new key when they change.
However, this usually is not enough for key rotation. Usually, you want
to proactively reencrypt all your fields with the new key, so that the
old key can be decommissioned.
This task allows you to do just that.

  
    
  
  Strategy


This task will migrate a table following this strategy:
	Page through the table using a cursor, selecting primary key values. By
default, the cursor is based on the schema's primary key. To use custom
fields in the cursor, see Cloak.CustomCursor.
Here's an example SQL query generated by mix cloak.migrate.ecto:
  SELECT primary_key
  FROM table_name
  WHERE primary_key > cursor
  ORDER BY primary_key ASC
  LIMIT 100

	For each primary key value, in a database transaction,
	Fetch the row with that key value, lock with "FOR UPDATE"
	Reencrypt all Cloak fields with the new cipher
	Write the row, unlocking it



Update queries are issued in parallel to maximize speed. Each row is fetched
and written back as quickly as possible to reduce the amount of time the
row is locked.

  
    
  
  Warnings


	mix cloak.migrate.ecto works well with primary keys of the following
 types. If your key is not of these types, see Cloak.CustomCursor.
	Integers
	PostgreSQL UUID
	MongoDB ObjectID


	Because mix cloak.migrate.ecto issues queries in parallel, it can consume
all your database connections. For this reason, you may wish to use a
separate Repo with a limited :pool just for Cloak migrations. This will
allow you to prevent any performance impact by throttling Cloak to use only
a limited number of database connections.



  
    
  
  Configuration


Ensure that you have configured your vault to use the new cipher by default!
# If using mix configuration...

config :my_app, MyApp.Vault,
  ciphers: [
    default: {Cloak.Ciphers.AES.GCM, tag: "NEW", key: <<...>>},
    retired: {Cloak.Ciphers.AES.CTR, tag: "OLD", key: <<...>>>}
  ]

# If configuring in the `init/1` callback:

defmodule MyApp.Vault do
  use Cloak.Vault, otp_app: :my_app

  @impl Cloak.Vault
  def init(config) do
    config =
      Keyword.put(config, :ciphers, [
        default: {Cloak.Ciphers.AES.GCM, tag: "NEW", key: <<...>>},
        retired: {Cloak.Ciphers.AES.CTR, tag: "OLD", key: <<...>>>}
      ])

    {:ok, config}
  end
end
If you want to migrate multiple schemas at once, you may find it convenient
to specify the schemas in your config/config.exs:
config :my_app,
  cloak_repo: [MyApp.Repo],
  cloak_schemas: [MyApp.Schema1, MyApp.Schema2]

  
    
  
  Usage


To run against only a specific repo and schema, use the -r and -s flags:
mix cloak.migrate.ecto -r MyApp.Repo -s MyApp.Schema
If you've configured multiple schemas at once, as shown above, you can simply
run:
mix cloak.migrate.ecto

